# Eliminate functions from system of partial differential equations

Here's the system of PDE

Eq1L = (-s^2)*β^2*up[x, y] + Derivative[0, 2][up][x, y] - β^2*Derivative[1, 0][θp][x, y] -
Derivative[1, 1][vp][x, y] + β^2*Derivative[1, 1][vp][x, y] +
β^2*Derivative[2, 0][up][x, y];
Eq2L = (-s^2)*β^2*vp[x, y] - β^2*Derivative[0, 1][θp][x, y] +
β^2*Derivative[0, 2][vp][x, y] - Derivative[1, 1][up][x, y] +
β^2*Derivative[1, 1][up][x, y] + Derivative[2, 0][vp][x, y];
Eq3L = s*ϵ1*ep[x, y] + s^2*ϵ1*τo*ep[x, y] - ϵ2*Qp[x] - s*ϵ2*τo*Qp[x] + s*θp[x, y] +
s^2*τo*θp[x, y] - Derivative[0, 2][θp][x, y] - Derivative[2, 0][θp][x, y];
Eq4L = -ep[x, y] + Derivative[0, 1][vp][x, y] + Derivative[1, 0][up][x, y];
Eq5L = β^2*ep[x, y] - β^2*θp[x, y] - σxp[x, y] - 2*Derivative[0, 1][vp][x, y];


How can eliminate up[x, y], vp[x, y] and ep[x, y] and obtain an equation of fourth order in θp[x, y] ?

• I do not receive any error messages from Eq3L // FullSimplify. Save your code and restart Mathematica to see what you then get. – bbgodfrey Sep 11 '16 at 3:24
• Also, no variables can be eliminated, because the question contains no equations. – bbgodfrey Sep 11 '16 at 3:35
• Thanks bbgodfrey. You mean you got a fourth or a 6th order differential equation in [Theta]p[x,y]?!. If So, would you please send or post the steps. – Essam Sep 11 '16 at 7:45

Again, this question is very similar to this one. And I think it's the original version of this question?

We first eliminate the ep[x, y] because there's no derivative of ep[x, y] in the system so eliminating the system first will make the analysis simpler:

neweq = List @@
Eliminate[{Eq1L == 0, Eq2L == 0, Eq3L == 0, Eq4L == 0, Eq5L == 0}, ep[x, y]] //
Simplify


Then eliminate up and vp:

Equal @@@ First@
Solve[Flatten[{neweq, D[neweq[[2]], x], D[neweq[[1]], {x, 2}], D[neweq[[1]], {y, 2}],
D[neweq[[3]], {x, 2}], D[neweq[[3]], {y, 2}]}], θp[x, y], {up[x, y],
vp[x, y], Derivative[0, 1][vp][x, y], Derivative[1, 0][up][x, y],
Derivative[1, 2][up][x, y],
Derivative[2, 1][vp][x, y], Derivative[3, 0][up][x, y],
Derivative[0, 3][vp][x, y]}] // Simplify

• Thanks xzczd. I will continue the solution by applying the Fourier exponential transform and then Fourier Cos and Sin transform. – Essam Dec 26 '16 at 18:08
• Thank you it is a good solution, I accept it. – Essam Sep 25 '18 at 17:39
• @Essam If you decide to accept this answer, you can click the checkmark sign. – xzczd Sep 26 '18 at 6:31